

## Ze-FmpGetMdl Ver.0.1

### Femap モデルファイルの読み込み例 Excel マクロ使用説明書

#### ソフトの概要

本資料は Femap API を使用した例として NASTRAN 板及び梁要素モデルの固有値解析結果を使用し、起動している Femap のモデルファイルの内容を Excel のワークシートに出力した説明書です。

API とはアプリケーション プログラミング インターフェイスことです。Visual Basic、VBA(Excel, Word, Access, ... )、C、C++、さらには Femap Basic Scripting Language から呼び出し可能な関数群です。

#### 出力例

- ①「物性値一覧表」シート : 物性値
- ②「板部材特性一覧表」シート : 板要素に使用する板厚
- ③「梁部材特性一覧表」シート : 梁要素に使用する部材特性
- ④「質量一覧表」シート : 節点質量
- ⑤「固有振動数一覧表」シート : 固有値計算結果の固有振動数
- ⑥「節点ベクトル一覧表」シート : 固有値計算結果の固有振動数と  
T1, T2, T3, R1, R2, R3

#### 注意事項

- ①本 VBA は個人所有ですが、参考にして改編を行ってもらって構いません。
- ②テストモデルでの機能のチェックは行っていますが、処理できないデータもあると思います。また、エラー処理等は十分に行っていません。  
使用者の責任において使用して下さい。
- ③不具合の修正、機能の追加については使用者が行って下さい。
- ④本 VBA の内容についての質問は受け付けません。ただし、費用が発生しても問題なければ質問、機能の追加または作成については受け付けます。

開発者: 茨木 栄      Mail: sakae-ibaraki@mbr.nifty.com

改正来歴

説明書 Rev.	システム Ver.	改正日	改正内容
0.0	0.1	2021/11/26	初版

## 目次

1. 概要 .....	1
2. 使用環境 .....	2
3. 使用例 .....	3
4. 検証 .....	8
5. まとめ .....	12
6. Excel VBA リスト .....	13

### 添付ファイル

#### ① 使用説明書

ZE-FmpGetMdl\_Ver. 0.1\_Femap モデルファイルの読み込み例 Excel マクロ  
使用説明書.pdf

#### ②VBA ファイル

ZE-FmpGetMdl\_Ver. 0.1. xlsx

#### ③テストファイル

- Femap モデルファイル  
¥TestData¥METH.modfem
- NASTRAN インプットデータ  
¥TestData¥METH000.dat
- NASTRAN アウトプットリスト  
¥TestData¥meth000.f06

## 1. 概要

本資料は、Excel VBA を使用し起動している Femap のモデルファイルを Femap API を使用して読み込み、ワークシートに出力した例をまとめたものです。

なお、モデルのデータ入出力以外に変形図、コンタ図等の図形の表示、ファイルへの保存、クリップボードへのコピー等も行えます。

API の詳細については Femap がインストールされているフォルダ内の以下のファイルを参照して下さい。日本語版は翻訳したものであるため英語版も参考にして下さい。また、参考プログラムもインストールされているフォルダ内の以下のフォルダに複数保存されています。

①日本語版説明書

インストールフォルダ¥pdf¥api. pdf

②英語版説明書

インストールフォルダ¥pdf¥EN¥api. pdf

③参考プログラム

インストールフォルダ¥API

注:本資料では起動している Femap のモデルファイル进行处理していますが、ファイル名を指定して Femap を起動し処理することも可能です。

## 2. 使用環境

使用した Excel VBA のバージョンと Femap のバージョンを以下に示す。




図 2-1 Excel VBA バージョン




図 2-2 Femap バージョン

### 3. 使用例

VBA が保存されている「ZE-FmpGetMdl\_Ver. 0.1.xlsm」を開くと図 3-1 が表示される。

対象とする Femap モデルファイルを表示し、各ボタンをクリックして Femap のモデルデータを読み込み、読み込み結果を各シートに出力する。出力結果の例を図 3-2～図 3-7 に示す。

解析結果に固有値計算結果があった場合は「最大次数の入力」ダイアログが表示されるので出力する最大の次数を入力して下さい。固有値計算結果の出力は処理時間が長くなるため必要最小限の次数を入力して下さい。

なお、図 3-2～図 3-7 に示すシートの内容は添付ファイルの「¥TestData¥METH.modfem」を Femap で表示して実行した結果です。




図 3-1 Excel 画面




図 3-2 「物性値一覧表」シート




図 3-3 「板部材特性一覧表」シート




図 3-4 「梁部材特性一覧表」シート




図 3-5 「質量一覧表」シート



次数	固有振動数
1	16.99059
2	25.18088
3	30.93025
4	34.5061
5	56.43985
6	70.62379
7	87.63762
8	113.1228
9	131.3812
10	148.5302

図 3-6 「固有振動数一覧表」シート

次数	周波数	ID	T1	T2	T3	R1	R2	R3
1	16.99059	1	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		2	1.951441E-01	-2.092586E-07	1.666323E-04	7.312487E-07	2.267685E-03	1.619287E-06
		3	6.788170E-01	-3.867170E-04	3.332633E-04	4.828731E-06	3.607197E-03	3.238534E-06
		4	1.302964E+00	-1.698172E-03	4.998918E-04	1.229358E-05	4.025651E-03	4.857802E-06
		5	1.921357E+00	-4.473709E-03	6.665164E-04	2.313191E-05	3.539186E-03	6.477069E-06
		6	2.391160E+00	-9.254251E-03	8.391360E-04	3.736181E-05	2.174058E-03	8.096336E-06
		7	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		8	1.941921E-01	-1.435076E-03	1.698363E-04	1.588520E-05	2.256638E-03	2.761667E-06
		9	6.755144E-01	-4.653105E-03	3.396714E-04	2.228872E-05	3.589711E-03	5.523333E-06
		10	1.296647E+00	-8.138118E-03	5.095039E-04	1.922841E-05	4.006301E-03	8.284939E-06
		11	1.912089E+00	-1.037855E-02	6.793324E-04	6.741770E-06	3.522467E-03	1.104667E-05
		12	2.379714E+00	-8.670298E-03	8.491558E-04	-1.511693E-05	2.164340E-03	1.360833E-05
		13	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		14	1.951086E-01	-1.675440E-04	-1.670774E-04	1.213593E-06	2.267261E-03	1.298263E-06
		15	6.786885E-01	-2.037809E-04	-3.341535E-04	-1.641127E-06	3.606453E-03	2.596527E-06

3次まで出力して途中の次数は省略しています。

46	-6.266721E-04	4.103172E-05	-2.400000E+00	3.507955E-03	-3.548101E-03	-1.399231E-07		
47	-1.020512E-03	-4.130508E-05	-1.232458E+00	2.310718E-03	-5.227569E-03	-1.044407E-06		
48	1.104201E-04	-1.678193E-04	-1.939004E+00	2.601400E-03	7.415977E-03	-4.129160E-07		
49	-9.640801E-05	-2.639892E-05	-3.564594E+00	4.714355E-03	4.938707E-03	1.200398E-07		
50	-3.318588E-04	2.306458E-05	-4.212285E+00	5.554542E-03	1.501139E-07	4.685509E-08		
51	-5.672990E-04	-2.928829E-06	-3.564661E+00	4.714364E-03	-4.938519E-03	-2.646148E-08		
52	-7.741138E-04	-1.209193E-04	-1.939077E+00	2.601378E-03	-7.416144E-03	5.058571E-07		
53	-9.327343E-05	-1.379976E-05	-2.298781E+00	-4.789471E-09	8.696510E-03	4.692404E-08		
54	-1.311992E-04	-2.077479E-06	-4.202267E+00	3.715374E-09	5.811383E-03	4.688243E-08		
55	-3.201628E-04	9.639975E-06	-4.961390E+00	1.883103E-09	1.623015E-07	4.683975E-08		
56	-5.091497E-04	2.133221E-05	-4.202334E+00	8.434187E-10	-5.811210E-03	4.679329E-08		
57	-5.471645E-04	3.301332E-05	-2.298855E+00	1.387968E-08	-8.696662E-03	4.678130E-08		
58	1.338897E-04	1.402211E-04	-1.939004E+00	-2.601367E-03	7.415995E-03	5.067457E-07		
59	-7.295931E-05	2.223952E-05	-3.564596E+00	-4.714348E-03	4.938704E-03	-2.628529E-08		
60	-3.084287E-04	-3.799231E-06	-4.212285E+00	-5.554543E-03	1.452960E-07	4.682388E-08		
61	-5.438850E-04	4.559542E-05	-3.564661E+00	-4.714371E-03	-4.938518E-03	1.200432E-07		
62	-7.506991E-04	1.869510E-04	-1.939081E+00	-2.601390E-03	-7.416118E-03	-4.123745E-07		

図 3-7 「節点ベクトル一覧表」シート

#### 4. 検証

表 4-1-1～表 4-6-2 に Femap モデルファイルから読み込んだ Excel の結果と NASTRAN のインプット及びアウトプットリストを示す。表 4-3-1 及び表 4-3-2 以外については同じ値となっている。

表 4-1-1 「物性値一覧表」シート

物性値一覧表

PID	タイトル	ヤング率	ポアソン比	密度
2	等方性 マテリアル	200000	0.3	7.85E-09
3	等方性 マテリアル	300000	0.3	7.85E-09

表 4-1-2 NASTRAN インプットデータ「MAT1」

```

$-----*-----*-----*-----*-----*-----*-----*-----*-----*-----*
$ Femap with NX Nastran Material 2 : 等方性 マテリアル
MAT1      2 200000.76923.08      .3 7.85-9      0.      0.
$ Femap with NX Nastran Material 3 : 等方性 マテリアル
MAT1      3 300000. 115385.      .3 7.85-9      0.      0.
$-----*-----*-----*-----*-----*-----*-----*-----*-----*-----*
    
```

表 4-2-1 「板部材特性一覧表」シート

板部材特性一覧表

PID	タイトル	MID	板厚
31	T=10	2	10
41	T=5	3	5

表 4-2-2 NASTRAN インプットデータ「PSHELL」

```

$-----*-----*-----*-----*-----*-----*-----*-----*-----*-----*
$ Femap with NX Nastran Property 31 : T=10
PSHELL    31      2      10.      2      2      0.+
+          0.
$ Femap with NX Nastran Property 41 : T=5
PSHELL    41      3      5.      3      3      0.+
+          0.
$-----*-----*-----*-----*-----*-----*-----*-----*-----*-----*
    
```

表 4-3-1 「梁部材特性一覧表」シート

梁部材特性一覧表																		
PID	タイトル	WID	A	Iz	Iy	J	NSW	C1	C2	D1	D2	E1	E2	F1	F2	K1	K2	I12
4	□50X30	3	1500	112500	312500	281948	0	-15	-25	15	-25	15	25	-15	25	1275.678	1274.962	0
5	◎50X2	3	442.9646	122811.9	122811.9	245575.7	0	0	-25	25	0	0	25	-25	0	236.5731	236.5731	0

注:K1 及び K2 は有効断面積です。NASTRAN インプットの断面積×K1 または K2 となります。

表 4-3-2 NASTRAN インプットデータ「PBAR」

```

$-----*-----*-----*-----*-----*-----*-----*-----*
$ Femap with NX Nastran Property 4 : □50X30
$ Femap with NX Nastran PropShape 4 : 1,0,50.,30.,0.,0.,0.,0.
$ Femap with NX Nastran PropOrient 4 : 1,0,0.,1.,2.,3.,4.,-1.,0.,0.
PBAR      4      3      1500. 112500. 312500. 281948.      0.      +
+         -15.   -25.   15.   -25.   15.   25.   -15.   25.+
+         .850452.8499747      0.
$ Femap with NX Nastran Property 5 : ◎50X2
$ Femap with NX Nastran PropShape 5 : 6,0,25.,0.,0.,0.,0.,3.
$ Femap with NX Nastran PropOrient 5 : 6,0,0.,1.,2.,3.,4.,-1.,0.,0.
PBAR      5      3442.9646 122812. 122812. 245576.      0.      +
+         0.   -25.   25.   0.   0.   25.   -25.   0.+
+         .5340677.5340677      0.
$-----*-----*-----*-----*-----*-----*-----*-----*
    
```

表 4-4-1 「質量一覧表」シート

質量一覧表

PID	タイトル	質量	オフセットX	オフセットY	オフセットZ
1	質量 フ°ロハ°テイ	0.01502	0	0	0
2	質量 フ°ロハ°テイ	0.01301	0	0	0

表 4-4-2 NASTRAN インプットデータ「CONM2」

```

$-----*-----*-----*-----*-----*-----*-----*-----*
CONM2      65      6      0 .01502      0.      0.      0.
CONM2      66     12      0 .01301      0.      0.      0.
$-----*-----*-----*-----*-----*-----*-----*
    
```

表 4-5-1 「固有振動数一覧表」シート

固有振動数一覧表

次数	周波数
1	16.99059
2	25.18088
3	30.93025
4	34.5061
5	56.43985
6	70.62379
7	87.63762
8	113.1228
9	131.3812
10	148.5302

表 4-5-2 NASTRAN アウトプットリスト「固有振動数」

MODE NO.	EXTRACTION ORDER	EIGENVALUE	REAL EIGENVALUES		GENERALIZED MASS	GENERALIZED STIFFNESS
			RADIANS	CYCLES		
1	1	1.139663E+04	1.067550E+02	1.699059E+01	1.000000E+00	1.139663E+04
2	2	2.503234E+04	1.582161E+02	2.518088E+01	1.000000E+00	2.503234E+04
3	3	3.776823E+04	1.943405E+02	3.093025E+01	1.000000E+00	3.776823E+04
4	4	4.700582E+04	2.168082E+02	3.450610E+01	1.000000E+00	4.700582E+04
5	5	1.257568E+05	3.546221E+02	5.643985E+01	1.000000E+00	1.257568E+05
6	6	1.969073E+05	4.437424E+02	7.062379E+01	1.000000E+00	1.969073E+05
7	7	3.032082E+05	5.506434E+02	8.763762E+01	1.000000E+00	3.032082E+05
8	8	5.051962E+05	7.107715E+02	1.131228E+02	1.000000E+00	5.051962E+05
9	9	6.814378E+05	8.254924E+02	1.313812E+02	1.000000E+00	6.814378E+05
10	10	8.709416E+05	9.332425E+02	1.485302E+02	1.000000E+00	8.709416E+05

表 4-6-1 「節点ベクトル一覧表」シート

節点ベクトル一覧表

次数	周波数	ID	T1	T2	T3	R1	R2	R3
1	16.99059	1	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		2	1.951441E-01	-2.092586E-07	1.666323E-04	7.312487E-07	2.267685E-03	1.619267E-06
		3	6.788170E-01	-3.867170E-04	3.332633E-04	4.828731E-06	3.607197E-03	3.238534E-06
		4	1.302964E+00	-1.698172E-03	4.998918E-04	1.229358E-05	4.025651E-03	4.857802E-06
		5	1.921357E+00	-4.473709E-03	6.665164E-04	2.313191E-05	3.539186E-03	6.477069E-06
		6	2.391160E+00	-9.254251E-03	8.331360E-04	3.736181E-05	2.174058E-03	8.096336E-06
		7	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		8	1.941921E-01	-1.435076E-03	1.698363E-04	1.588520E-05	2.256638E-03	2.761667E-06
		9	6.755144E-01	-4.653105E-03	3.396714E-04	2.228872E-05	3.589711E-03	5.523333E-06
		10	1.296647E+00	-8.138118E-03	5.095039E-04	1.922841E-05	4.006301E-03	8.284999E-06
		11	1.912089E+00	-1.037855E-02	6.793324E-04	6.741770E-06	3.522467E-03	1.104667E-05
		12	2.379714E+00	-9.870298E-03	8.491558E-04	-1.511693E-05	2.164340E-03	1.380833E-05
		13	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00	0.000000E+00
		14	1.951086E-01	-1.675440E-04	-1.670774E-04	1.213593E-06	2.267261E-03	1.298263E-06
		15	6.786865E-01	-2.037809E-04	-3.341535E-04	-1.641127E-06	3.606453E-03	2.596527E-06

表 4-6-2 NASTRAN アウトプットリスト「節点ベクトル」

EIGENVALUE = 1.139663E+04								
CYCLES = 1.699059E+01								
REAL EIGENVECTOR NO. 1								
POINT ID.	TYPE	T1	T2	T3	R1	R2	R3	
1	G	0.0	0.0	0.0	0.0	0.0	0.0	0.0
2	G	1.951441E-01	-2.092586E-07	1.666323E-04	7.312487E-07	2.267685E-03	1.619267E-06	
3	G	6.788170E-01	-3.867170E-04	3.332633E-04	4.828731E-06	3.607197E-03	3.238534E-06	
4	G	1.302964E+00	-1.698172E-03	4.998918E-04	1.229358E-05	4.025651E-03	4.857802E-06	
5	G	1.921357E+00	-4.473709E-03	6.665164E-04	2.313191E-05	3.539186E-03	6.477069E-06	
6	G	2.391160E+00	-9.254251E-03	8.331360E-04	3.736181E-05	2.174058E-03	8.096336E-06	
7	G	0.0	0.0	0.0	0.0	0.0	0.0	
8	G	1.941921E-01	-1.435076E-03	1.698363E-04	1.588520E-05	2.256638E-03	2.761667E-06	
9	G	6.755144E-01	-4.653105E-03	3.396714E-04	2.228872E-05	3.589711E-03	5.523333E-06	
10	G	1.296647E+00	-8.138118E-03	5.095039E-04	1.922841E-05	4.006301E-03	8.284999E-06	
11	G	1.912089E+00	-1.037855E-02	6.793324E-04	6.741770E-06	3.522467E-03	1.104667E-05	
12	G	2.379714E+00	-9.870298E-03	8.491558E-04	-1.511693E-05	2.164340E-03	1.380833E-05	
13	G	0.0	0.0	0.0	0.0	0.0	0.0	
14	G	1.951086E-01	-1.675440E-04	-1.670774E-04	1.213593E-06	2.267261E-03	1.298263E-06	
15	G	6.786865E-01	-2.037809E-04	-3.341535E-04	-1.641127E-06	3.606453E-03	2.596527E-06	
以下省略								

## 5. まとめ

本使用例では Femap モデルファイルの読み込みしか行っていないが書き込みも可能です。読み込みについては必要なデータのみ処理すればよいが書き込みについては必要なデータはすべて書き込む必要がある。なお、解析モデルをジオメトリから作成することも可能です。

どのデータが必要かについては説明書には記載がないため、一度モデルデータを読み込んでデータの内容を確認する必要がある。

また、モデルファイルの解析結果を処理する場合はデータ量が多いため Excel VBA では処理に時間がかかる。また、Excel2007 以降のシートの最大行は 1048576 でありシートに出力できない場合がある。

使用した感じでは、Visual Studio 等の実行ファイルを作成できる言語スプレッドシートを使用した方がよい。

Femap の「ツール」→「プログラミング」→「API プログラミング」メニューで使用できる Visual Basic もあるがエディタが非常に使いにくい。

## 6. Excel VBA リスト

Excel VBA で Femap API を使用するには VBA エディタの「ツール」→「参照設定」メニューをクリックし図 6-1 に示す「参照設定」ダイアログを表示し Femap のライブラリを参照する必要があります。

表 6-1 に本 VBA の関数一覧表を、表 6-2～表 6-18 に本資料で使用した VBA のリストを示す。




図 6-1 「参照設定」ダイアログ



表 6-1 関数一覧表

No.	Function Name	内容
1		定数
2	MainAll	全データを読み込みシートに出力
3	FeMaterial	Femap モデルファイルの MATERIAL データを「物性値一覧表」シートに出力
4	InitMaterial	「物性値一覧表」シートの作成
5	FePropatyShell	Femap モデルファイルの PROPATY(板)データを「板部材特性一覧表」シートに出力
6	InitPshell	「板部材特性一覧表」シートの作成
7	FePropatyBar	Femap モデルファイルの PROPATY(梁)データを「梁部材特性一覧表」シートに出力
8	InitPbar	「梁部材特性一覧表」シートの作成
9	FePropatyMass	Femap モデルファイルの PROPATY(質量)データを「質量特性一覧表」シートに出力
10	InitPmass	「質量特性一覧表」シートの作成
11	FeCycle	Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数を「固有振動数一覧表」シートに出力
12	InitCycle	「固有振動数一覧表」シートの作成
13	FeFrequency	Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、T1, T2, T3, R1, R2, R3 を「節点ベクトル一覧表」シートに出力
14	InitFrequency	「節点ベクトル一覧表」シートの作成
15	Help	使用説明書の表示
16	CellsSet	左上と右下の Row, Col を指定してセル書式設定
17	BoderRange	左上と右下の Row, Col を指定して罫線の設定

表 6-2 定数

```

' *****
' Femap モデルデータを読み込み各シートに出力
' 作成 2021/07/31 茨木 栄 (jah01272@nifty.com)
' 注:Femap でモデルを表示している必要がある。
' *****
Option Explicit

' シート名
Public Const SHEETNAME_MAIN = "メイン"
Public Const SHEETNAME_MAT = "物性値一覧表"
Public Const SHEETNAME_PHELL = "板部材特性一覧表"
Public Const SHEETNAME_PBAR = "梁部材特性一覧表"
Public Const SHEETNAME_PMASS = "質量一覧表"
Public Const SHEETNAME_CYCLE = "固有振動数一覧表"
Public Const SHEETNAME_FREQUENCY = "節点ベクトル一覧表"

' フォント
Const FONT_NAME = "MS ゴシック"
Const FONT_SIZE = 10

' 指数書式
Const UNIT_FORM_OUTPUT = "0.000000E+00"

' Excel の最大行数
Const EXCEL_MAX_ROW = 1048576
    
```

表 6-3(1/2) 全データを読み込みシートに出力

```
'=====
'全データを読み込みシートに出力
'=====

Function MainAll() As String
    Dim objSheet As Object
    Dim errMsg As String ' エラーメッセージ

    Application.DisplayAlerts = False ' 警告を出さない
    Application.ScreenUpdating = False ' 画面の更新を停止する
    Application.Cursor = xlWait

    'ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
    For Each objSheet In ThisWorkbook.Worksheets
        If objSheet.Name <> SHEETNAME_MAIN Then '「メイン」シート以外
            Worksheets(objSheet.Name).Delete ' 既存シートの削除
        End If
    Next

    ' Femap モデルファイルの MATERIAL データをシートに出力
    errMsg = FeMaterial(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If

    ' Femap モデルファイルの PROPATY (板) データをシートに出力
    errMsg = FePropatyShell(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If

    ' Femap モデルファイルの PROPATY (質量) データをシートに出力
    errMsg = FePropatyBar(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If

    ' Femap モデルファイルの PROPATY (梁) データをシートに出力
    errMsg = FePropatyMass(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If

    ' Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数シートに出力
    errMsg = FeCycle(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If

    ' Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、T1, T2, T3, R1, R2, R3 をシートに出力
    errMsg = FeFrequency(True)
    If Len(errMsg) <> 0 Then
        GoTo myError
    End If
```

表 6-3(2/2) 全データを読み込みシートに出力

```
Application.Cursor = xlDefault
Application.ScreenUpdating = True '画面を更新する
Application.DisplayAlerts = True '警告を出す

MsgBox "全データシートの作成を終了しました。", vbOKOnly, "終了"
Exit Function

myError:
Application.Cursor = xlDefault
Application.ScreenUpdating = True '画面を更新する
Application.DisplayAlerts = True '警告を出す
MsgBox errMsg, vbOKOnly, "警告"
End Function
```

表 6-4(1/2) Femap モデルファイルの MATERIAL データを「物性値一覧表」シートに出力

```

' =====
' Femap モデルファイルの MATERIAL データを「物性値一覧表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FeMaterial(Optional Flag As Boolean = False) As String
    Dim objSheet As Object
    Dim errMsg As String ' エラーメッセージ

    Dim femap As Object ' Femap
    Dim Row As Long ' 行番号

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    ' Material オブジェクトを生成
    Dim mt As Object
    Set mt = femap.feMatl()

    ' 物性値があるかチェック
    If mt.First = 0 Then ' 物性値が無かった
        If Flag Then
            FeMaterial = ""
        Else
            MsgBox "物性値データはありませんでした。", vbOKOnly, "終了"
        End If
        Exit Function
    End If

    ' =====
    ' 「物性値一覧表」シートの追加
    ' =====

    errMsg = InitMaterial

    Row = 3 ' 開始行

    mt.First
    Do
        Cells(Row, 1).Value = mt.id ' ID
        Cells(Row, 2).Value = mt.Title ' タイトル
        Cells(Row, 3).Value = mt.Ex ' ヤング率
        Cells(Row, 4).Value = mt.Nuxy ' ポアソン比
        Cells(Row, 5).Value = mt.Density ' 密度
        Row = Row + 1
    Loop While mt.Next
    Row = Row - 1

    Cells(1, 1).Select ' セルの移動

```

表 6-4(2/2) Femap モデルファイルの MATERIAL データを「物性値一覧表」シートに出力

```

'-----
'罫線の作成及びウインドウ枠の固定
'-----
errMsg = BoderRange(2, 1, Row, 5) '罫線

If Not Flag Then '単独処理の場合
    Application.Cursor = xlDefault
    Application.ScreenUpdating = True '画面を更新する
    Application.DisplayAlerts = True '警告を出す
    MsgBox "[" & SHEETNAME_MAT & "]"シートの作成を終了しました。", vbOKOnly, "終了"
End If

FeMaterial = ""
Exit Function

myError:
If Not Flag Then '単独処理の場合
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"
Else
    FeMaterial = "Femap モデルファイルを開いて下さい。"
End If
End Function

```

表 6-5 「物性値一覧表」シートの作成」シートに出力

```

' =====
' 「物性値一覧表」シートの作成
' =====
Function InitMaterial() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("PID", "タイトル", "ヤング率", "ポアソン比", "密度")

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_MAT ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 10
    Columns(1).ColumnWidth = 8
    Columns(2).ColumnWidth = 20

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Columns(2).HorizontalAlignment = xlLeft

    Cells(1, 1).Value = SHEETNAME_MAT ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```

表 6-6(1/2) Femap モデルファイルの PROPATY (板) データを「板部材特性一覧表」シートに出力

```

' =====
' Femap モデルファイルの PROPATY (板) データを「板部材特性一覧表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FePropatyShell(Optional Flag As Boolean = False) As String
    Dim errMsg As String ' エラーメッセージ
    Dim objSheet As Object

    Dim femap As Object ' Femap
    Dim Row As Long ' 行番号

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    ' Propaty オブジェクトを生成
    Dim pr As Object
    Set pr = femap.feProp()

    ' 板要素プロパティがあるかチェック
    Dim PropFlag As Boolean ' プロパティ有無のフラッグ
    pr.First
    Do
        If pr.Type = 17 Then
            PropFlag = True
            Exit Do
        End If
    Loop While pr.Next

    If Not PropFlag Then ' プロパティが無かった
        If Flag Then
            FePropatyShell = ""
        Else
            MsgBox "板要素プロパティはありませんでした。", vbOKOnly, "終了"
        End If
        Exit Function
    End If

```



表 6-6(2/2) Femap モデルファイルの PROPATY (板) データを「板部材特性一覧表」シートに出力

```

'-----
'「板部材特性一覧表」シートの追加
'-----
errMsg = InitPshell

Row = 3 ' 開始行

pr.First
Do
  If pr.Type = 17 Then
    Cells(Row, 1).Value = pr.id      ' ID
    Cells(Row, 2).Value = pr.Title  ' タイトル
    Cells(Row, 3).Value = pr.matIID ' MID
    Cells(Row, 4).Value = pr.pval(0) ' 板厚
    Row = Row + 1
  End If
Loop While pr.Next
Row = Row - 1

Cells(1, 1).Select ' セルの移動

'-----
'罫線の作成及びウインドウ枠の固定
'-----
errMsg = BoderRange(2, 1, Row, 4) ' 罫線

If Not Flag Then ' 単独処理の場合
  Application.Cursor = xlDefault
  Application.ScreenUpdating = True ' 画面を更新する
  Application.DisplayAlerts = True ' 警告を出す
  MsgBox "[F" & SHEETNAME_P SHELL & "]"シートの作成を終了しました。", vbOKOnly, "終了"
End If

FePropatyShell = ""
Exit Function

myError:
If Not Flag Then ' 単独処理の場合
  MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"
Else
  FePropatyShell = "Femap モデルファイルを開いて下さい。"
End If
End Function

```

表 6-7 「板部材特性一覧表」シートの作成

```

'
=====
' 「板部材特性一覧表」シートの作成
=====
Function InitPshell() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_PSHELL ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("PID", "タイトル", "MID", "板厚")

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 10
    Columns(1).ColumnWidth = 8
    Columns(2).ColumnWidth = 20

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Columns(2).HorizontalAlignment = xlLeft

    Cells(1, 1).Value = SHEETNAME_PSHELL ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```

表 6-8(1/2) Femap モデルファイルの PROPATY (梁) データを「梁部材特性一覧表」シートに出力

```

' =====
' Femap モデルファイルの PROPATY (梁) データを「梁部材特性一覧表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FePropatyBar(Optional Flag As Boolean = False) As String
    Dim errMsg As String ' エラーメッセージ
    Dim objSheet As Object

    Dim femap As Object ' Femap
    Dim Row As Long ' 行番号

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    ' Propaty オブジェクトを生成
    Dim pr As Object
    Set pr = femap.feProp()

    ' 梁要素プロパティがあるかチェック
    Dim PropFlag As Boolean ' プロパティ有無のフラッグ
    pr.First
    Do
        If pr.Type = 2 Then ' バー
            If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Or pr.flagI(1) = 5 Or pr.flagI(1) = 6 Then ' 形状入力は処理しない
                PropFlag = True
                Exit Do
            End If
        End If
    Loop While pr.Next

    If Not PropFlag Then ' プロパティが無かった
        If Flag Then
            FePropatyBar = ""
        Else
            MsgBox "梁要素プロパティはありませんでした。", vbOKOnly, "終了"
        End If
        Exit Function
    End If

```

表 6-8(2/2) Femap モデルファイルの PROPATY(梁)データを「梁部材特性一覧表」シートに出力

```

'-----
' 梁部材特性一覧表」シートの追加
'-----
errMsg = InitPbar

Row = 3 ' 開始行

pr.First
Do
  If pr.Type = 2 Then ' バー
    If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Or pr.flagI(1) = 5 Or pr.flagI(1) = 6 Then ' 形状入力処理しない
      Cells(Row, 1).Value = pr.id ' ID
      Cells(Row, 2).Value = pr.Title ' タイトル
      Cells(Row, 3).Value = pr.matID ' MID
      Cells(Row, 4).Value = pr.pval(0) ' A
      Cells(Row, 5).Value = pr.pval(1) ' Iz
      Cells(Row, 6).Value = pr.pval(2) ' Iy
      Cells(Row, 7).Value = pr.pval(4) ' J
      Cells(Row, 8).Value = pr.pval(7) ' NSM
      Cells(Row, 9).Value = pr.pval(8) ' C1
      Cells(Row, 10).Value = pr.pval(9) ' C2
      Cells(Row, 11).Value = pr.pval(10) ' D1
      Cells(Row, 12).Value = pr.pval(11) ' D2
      Cells(Row, 13).Value = pr.pval(12) ' E1
      Cells(Row, 14).Value = pr.pval(13) ' E2
      Cells(Row, 15).Value = pr.pval(14) ' F1
      Cells(Row, 16).Value = pr.pval(15) ' F2
      Cells(Row, 17).Value = pr.pval(5) ' K1
      Cells(Row, 18).Value = pr.pval(6) ' K2
      Cells(Row, 19).Value = pr.pval(3) ' I12
      Row = Row + 1
    End If
  End If
Loop While pr.Next
Row = Row - 1

Cells(1, 1).Select ' セルの移動

'-----
' 罫線の作成及びウインドウ枠の固定
'-----
errMsg = BoderRange(2, 1, Row, 19) ' 罫線

If Not Flag Then ' 単独処理の場合
  Application.Cursor = xlDefault
  Application.ScreenUpdating = True ' 画面を更新する
  Application.DisplayAlerts = True ' 警告を出す
  MsgBox "[I" & SHEETNAME_PBAR & "]シートの作成を終了しました。", vbOKOnly, "終了"
End If

FePropatyBar = ""
Exit Function

myError:
If Not Flag Then ' 単独処理の場合
  MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"
Else
  FePropatyBar = "Femap モデルファイルを開いて下さい。"
End If
End Function

```

表 6-9 「梁部材特性一覧表」シートの作成

```

' =====
' 「梁部材特性一覧表」シートの作成
' =====
Function InitPbar() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("PID", "タイトル", "MID", "A", "Iz", "Iy", "J", "NSM", "C1", "C2", "D1", "D2", "E1", "E2", "F1",
    "F2", "K1", "K2", "I12")

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_PBAR ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 10
    Columns(1).ColumnWidth = 8
    Columns(2).ColumnWidth = 20

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Columns(2).HorizontalAlignment = xlLeft

    Cells(1, 1).Value = SHEETNAME_PBAR ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での
    中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```

表 6-10(1/2) Femap モデルファイルの PROPATY (質量) データを「質量特性一覧表」シートに出力

```

' =====
' Femap モデルファイルの PROPATY (質量) データを「質量特性一覧表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FePropatyMass(Optional Flag As Boolean = False) As String
    Dim errMsg As String ' エラーメッセージ
    Dim objSheet As Object

    Dim femap As Object ' Femap
    Dim Row As Long ' 行番号

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    ' Propaty オブジェクトを生成
    Dim pr As Object
    Set pr = femap.feProp()

    ' 質量要素プロパティがあるかチェック
    Dim PropFlag As Boolean ' プロパティ有無のフラッグ
    pr.First
    Do
        If pr.Type = 27 Then ' 質量
            PropFlag = True
            Exit Do
        End If
    Loop While pr.Next

    If Not PropFlag Then ' プロパティが無かった
        If Flag Then
            FePropatyMass = ""
        Else
            MsgBox "質量要素プロパティはありませんでした。", vbOKOnly, "終了"
        End If
        Exit Function
    End If

```

表 6-10(2/2) Femap モデルファイルの PROPATY(質量)データを「質量特性一覧表」シートに出力

```

'-----
'「質量特性一覧表」シートの追加
'-----
errMsg = InitPmass

Row = 3 '開始行

pr.First
Do
  If pr.Type = 27 Then '質量
    If pr.flagI(1) = 0 Or pr.flagI(1) = 1 Then '形状入力は処理しない
      Cells(Row, 1).Value = pr.id ' ID
      Cells(Row, 2).Value = pr.Title ' タイトル
      Cells(Row, 3).Value = pr.pval(7) ' 質量
      Cells(Row, 4).Value = pr.pval(8) ' オフセット X
      Cells(Row, 5).Value = pr.pval(9) ' オフセット Y
      Cells(Row, 6).Value = pr.pval(10) ' オフセット Z
      Row = Row + 1
    End If
  End If
Loop While pr.Next
Row = Row - 1

Cells(1, 1).Select 'セルの移動

'-----
'罫線の作成及びウインドウ枠の固定
'-----
errMsg = BoderRange(2, 1, Row, 6) '罫線

If Not Flag Then '単独処理の場合
  Application.Cursor = xlDefault
  Application.ScreenUpdating = True '画面を更新する
  Application.DisplayAlerts = True '警告を出す
  MsgBox "「I” & SHEETNAME_PMASS & ”J”シートの作成を終了しました。", vbOKOnly, "終了"
End If

FePropatyMass = ""
Exit Function

myError:
If Not Flag Then '単独処理の場合
  MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"
Else
  FePropatyMass = "Femap モデルファイルを開いて下さい。"
End If
End Function

```

表 6-11 「質量特性一覧表」シートの作成

```

'=====
'「質量特性一覧表」シートの作成
'=====
Function InitPmass() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("PID", "タイトル", "質量", "オフセット X", "オフセット Y", "オフセット Z")

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_PMASS ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 10
    Columns(1).ColumnWidth = 8
    Columns(2).ColumnWidth = 20

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Columns(2).HorizontalAlignment = xlLeft

    Cells(1, 1).Value = SHEETNAME_PMASS ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```



表 6-12(1/2) Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数を  
「固有振動数一覧表」シートに出力

```

' =====
' Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数を「固有振動数一覧表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FeCycle(Optional Flag As Boolean = False) As String
    Dim errMsg As String ' エラーメッセージ
    Dim objSheet As Object

    Dim femap As Object ' Femap
    Dim Row As Long ' 行番号

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    ' OutputSet オブジェクトを生成
    Dim Os As Object
    Set Os = femap.feOutputSet

    ' 固有値計算結果があるかチェック
    If Os.First = 0 Then ' 計算結果が無かった
        If Flag Then
            FeCycle = ""
        Else
            MsgBox "固有値計算結果はありませんでした。", vbOKOnly, "終了"
        End If
        Exit Function
    End If

    ' =====
    ' 「固有振動数一覧表」シートの追加
    ' =====

    errMsg = InitCycle

    Dim Mode As Long

    Row = 3 ' 開始行

    Os.First
    Do
        Mode = Mode + 1
        Cells(Row, 1).Value = Mode ' 次数
        Cells(Row, 2).Value = Os.Value ' 周波数
        Row = Row + 1
    Loop While Os.Next
    Row = Row - 1

    Cells(1, 1).Select ' セルの移動

```

表 6-12(2/2) Femap モデルファイルの OutputSet データから固有値計算結果の次数と周波数を  
「固有振動数一覧表」シートに出力

```

' -----
' 罫線の作成及びウインドウ枠の固定
' -----
errMsg = BoderRange(2, 1, Row, 2) ' 罫線

If Not Flag Then ' 単独処理の場合
    Application.Cursor = xlDefault
    Application.ScreenUpdating = True ' 画面を更新する
    Application.DisplayAlerts = True ' 警告を出す
    MsgBox "[" & SHEETNAME_CYCLE & "]"シートの作成を終了しました。", vbOKOnly, "終了"
End If

FeCycle = ""
Exit Function

myError:
If Not Flag Then ' 単独処理の場合
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"
Else
    FeCycle = "Femap モデルファイルを開いて下さい。"
End If
End Function

```

表 6-13 「固有振動数一覧表」シートの作成

```

' =====
' 「固有振動数一覧表」シートの作成
' =====
Function InitCycle() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("次数", "周波数")

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_CYCLE ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 10
    Columns(1).ColumnWidth = 5

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Cells(1, 1).Value = SHEETNAME_CYCLE ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```

表 6-14(1/3) Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、  
T1, T2, T3, R1, R2, R3 を「節点ベクトル一覧表」シートに出力

```

' =====
' Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、T1, T2, T3, R1, R2, R3 を「節点ベクトル一覧
' 表」シートに出力
' Flag : 全データの処理の場合は True
' =====
Function FeFrequency(Optional Flag As Boolean = False) As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ
    Dim objSheet As Object

    Dim femap As Object ' Femap

    Dim Mode As Long      ' 次数
    Dim Row As Long      ' 行番号
    Dim Tr As Object ' T1, T2, T3 の読み込み結果
    Dim Ro As Object ' R1, R2, R3 の読み込み結果

    ' 出力データ
    Dim tCount As Long ' 読み込み数 (T1, T2, T3)
    Dim rCount As Long ' 読み込み数 (R1, R2, R3)
    Dim TrId As Variant ' ID (T1, T2, T3) の読み込み結果
    Dim RoId As Variant ' ID (R1, R2, R3) の読み込み結果
    Dim t1 As Variant ' T1 の読み込み結果
    Dim t2 As Variant ' T2 の読み込み結果
    Dim t3 As Variant ' T3 の読み込み結果
    Dim r1 As Variant ' R1 の読み込み結果
    Dim r2 As Variant ' R2 の読み込み結果
    Dim r3 As Variant ' R3 の読み込み結果

    Dim rc1, rc2 As Integer ' Femap 読み込みのリターンコード

    If Not Flag Then ' 単独処理の場合
        Application.DisplayAlerts = False ' 警告を出さない
        Application.ScreenUpdating = False ' 画面の更新を停止する
        Application.Cursor = xlWait

        ' ブックの全シートを 1 つずつループして「メイン」シート以外を削除する
        For Each objSheet In ThisWorkbook.Worksheets
            If objSheet.Name <> SHEETNAME_MAIN Then ' 「メイン」シート以外
                Worksheets(objSheet.Name).Delete ' 既存シートの削除
            End If
        Next
    End If

    ' 処理する最大次数の入力
    Dim MaxMode As Integer
    Do
        MaxMode = Val(StrConv(InputBox("処理する最大次数を入力して下さい。", "最大次数の入力", 1), vbNarrow))
        If MaxMode >= 0 Then Exit Do
    Loop
    If MaxMode = 0 Then
        Exit Function
    End If

    On Error GoTo myError

    ' すでに実行されている FEMAP に接続
    Set femap = GetObject(, "femap.model")
    On Error GoTo 0

    Dim Os As Object
    Set Os = femap.feOutputSet

```

表 6-14(2/3) Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、  
T1, T2, T3, R1, R2, R3 を「節点ベクトル一覧表」シートに出力

```

'固有値計算結果があるかチェック
If Os.First = 0 Then '計算結果が無かった
  If Flag Then
    FeFrequency = ""
  Else
    MsgBox "固有値計算結果はありませんでした。", vbOKOnly, "終了"
  End If
  Exit Function
End If

'-----
'「節点ベクトル一覧表」シートの追加
'-----

errMsg = InitFrequency

Row = 3 '開始行

Mode = 0 '次数

Os.First
Do

  'Load the Translation Data Vector
  Set Tr = Os.vector(1)

  'Load the Rotation Data Vector
  Set Ro = Os.vector(5)

  '全計算結果の読み込み
  rc1 = Tr.GetVectorAtNode(tCount, TrId, t1, t2, t3) 'T1, T2, T3 の読み込み
  rc2 = Ro.GetVectorAtNode(rCount, RoId, r1, r2, r3) 'R1, R2, R3 の読み込み

  Mode = Mode + 1

  '次数と周波数を全節点に出力すると処理時間が長くなるため最初の節点番号のみ出力
  Cells(Row, 1).Value = Mode '次数
  Cells(Row, 2).Value = Os.Value '周波数

  For i = 0 To tCount - 1
    Cells(Row, 3).Value = TrId(i) 'ID
    If rc1 = -1 Then
      Cells(Row, 4).Value = t1(i) 'T1
      Cells(Row, 5).Value = t2(i) 'T2
      Cells(Row, 6).Value = t3(i) 'T3
    End If
    If rc2 = -1 Then
      Cells(Row, 7).Value = r1(i) 'R1
      Cells(Row, 8).Value = r2(i) 'R2
      Cells(Row, 9).Value = r3(i) 'R3
    End If
    Row = Row + 1

    'Excel の制限の行数を超えた場合は終了
    If Row > EXCEL_MAX_ROW Then
      Exit Do
    End If

  Next
  If Mode = MaxMode Then Exit Do '最大次数になった場合は終了
Loop While Os.Next
Row = Row - 1

Cells(1, 1).Select 'セルの移動

```

表 6-14(3/3) Femap モデルファイルの OutputSet データから固有値計算結果の次数、周波数、T1, T2, T3, R1, R2, R3 を「節点ベクトル一覧表」シートに出力

```
'-----  
' 罫線の作成及びウインドウ枠の固定  
'-----  
errMsg = BoderRange(2, 1, Row, 9) ' 罫線  
  
If Not Flag Then ' 単独処理の場合  
    Application.Cursor = xlDefault  
    Application.ScreenUpdating = True ' 画面を更新する  
    Application.DisplayAlerts = True ' 警告を出す  
    If Row <> EXCEL_MAX_ROW Then  
        MsgBox "I" & SHEETNAME_FREQUENCY & "Jシートで Excel 制限の行数を超えました。", vbOKOnly, "警告"  
    Else  
        MsgBox "I" & SHEETNAME_FREQUENCY & "Jシートの作成を終了しました。", vbOKOnly, "終了"  
    End If  
Else  
    If Row <> EXCEL_MAX_ROW Then  
        FeFrequency = ""  
    Else  
        FeFrequency = "I" & SHEETNAME_FREQUENCY & "Jシートで Excel 制限の行数を超えました。"  
    End If  
End If  
  
Exit Function  
  
myError:  
If Not Flag Then ' 単独処理の場合  
    MsgBox "Femap モデルファイルを開いて下さい。", vbOKOnly, "警告"  
Else  
    FeFrequency = "Femap モデルファイルを開いて下さい。"  
End If  
End Function
```

表 6-15 「節点ベクトル一覧表」シートの作成

```

' =====
' 「節点ベクトル一覧表」シートの作成
' =====
Function InitFrequency() As String
    Dim i As Long
    Dim errMsg As String ' エラーメッセージ

    ' 表のヘッダー
    Dim Head As Variant
    Head = Array("次数", "周波数", "ID", "T1", "T2", "T3", "R1", "R2", "R3")

    Worksheets.Add(After:=Worksheets(Worksheets.Count), Count:=1).Name = SHEETNAME_FREQUENCY ' シートの追加
    ActiveSheet.Tab.Color = RGB(0, 255, 0) ' タブカラー

    ' フォント設定
    Cells.Select
    With Selection.Font
        .Name = FONT_NAME
        .Size = FONT_SIZE
    End With

    ' セル高さ
    Cells.RowHeight = 20
    Rows(1).RowHeight = 30

    ' セル幅
    Cells.ColumnWidth = 12.5
    Columns(1).ColumnWidth = 5
    Columns(2).ColumnWidth = 10
    Columns(3).ColumnWidth = 10

    ' ベクトルのセル書式
    Range(Colomns(4), Colomns(UBound(Head) + 1)).Select
    Selection.NumberFormatLocal = UNIT_FORM_OUTPUT

    ' 列幅の設定
    For i = 0 To UBound(Head)
        Cells(2, i + 1).Value = Head(i)
    Next

    Cells(1, 1).Value = SHEETNAME_FREQUENCY ' タイトル

    ' タイトルの書式
    errMsg = CellsSet(1, 1, 1, UBound(Head) + 1, xlCenterAcrossSelection, False, False) ' タイトル 選択範囲内での
    中央

    ' ヘッダーの書式
    errMsg = CellsSet(2, 1, 2, UBound(Head) + 1, xlCenter, False, False) ' 表のヘッダー選択範囲での中央

End Function

```

表 6-16 使用説明書の表示

```

' =====
' 使用説明書の表示
' =====
Function Help()
    CreateObject("Shell.Application").ShellExecute ActiveWorkbook.Path & "\*" & "ZE-FmpGetMdl_Ver. 0.1_Femap モデルデータの読み込み例 Excel マクロ使用説明書.pdf"
End Function
    
```

表 6-17 左上と右下の Row, Col を指定してセル書式設定

```

' =====
' 左上と右下の Row, Col を指定してセル書式設定
' Row1          : 範囲開始行 1
' Col1          : 範囲開始列 1
' Row2          : 範囲開始行 2 行 1 より小さければ列 1
' Col2          : 範囲開始列 2 列 1 より小さければ列 1
' CellAlignment: 水平の書式
' FlagMerge     : マージフラッグ
' FlagWrap      : 文字の折り返しフラッグ
' =====
Function CellsSet(ByVal Row1 As Long, ByVal Col1 As Long, ByVal Row2 As Long, ByVal Col2 As Long, ByRef CellAlignment As Integer, ByRef FlagMerge As Boolean, ByRef FlagWrap As Boolean) As String
    ' 範囲行列の設定
    If Row1 < 0 Then
        Row1 = 1
    End If
    If Col1 < 0 Then
        Col1 = 1
    End If
    If Row2 < Row1 Then
        Row2 = Row1
    End If
    If Col2 < Col1 Then
        Col2 = Col1
    End If

    ' セル範囲
    Range(Cells(Row1, Col1), Cells(Row2, Col2)).Select

    ' 書式の設定
    With Selection
        .HorizontalAlignment = CellAlignment ' 横位置
        .VerticalAlignment = xlCenter        ' 縦位置
        .WrapText = FlagWrap                ' 文字の折り返し
        .Orientation = 0                    ' 文字列の向き
        .AddIndent = False                  ' 前後にスペース
        .IndentLevel = 0                    ' インデント
        .ShrinkToFit = False                ' 縮小して全体表示
        .ReadingOrder = xlContext           ' 文字列の並び(日本語には無意味)
        .MergeCells = FlagMerge              ' セルの結合
    End With
End Function
    
```

表 6-18 左上と右下の Row, Col を指定して罫線の設定

```

' =====
' 左上と右下の Row, Col を指定して罫線の設定
' =====
Function BoderRange(Row1, Col1, Row2, Col2) As String
    Range(Cells(Row1, Col1), Cells(Row2, Col2)).Borders.LineStyle = xlContinuous
End Function
    
```